Num. Math.—12

---

# Gauss kvadratúra

Általánosított kvadratúra probléma: $\int_a^b f[x]\ \omega\ [x]\ dx$

Most csak azzal foglakozunk, amikor $\omega$=1, [a,b]=[-1,1].

Nem ekvidisztáns alappontrendszer, n pont esetén [-1,1]-en minden(!) (2n-1)-ed fokú polinomra pontos.

n=2, 3-adfokú
n=3, 5-ödfokú stb.

Hogyan kapjuk a pontokat és a súlyokat? $\int_a^b f[x]\ dx = \sum_{j=0}^{n-1} f\left[x_j\right] w_j$

Alappontok: Ortogonális polinomrendszer elemeinek gyökei
Súlyok: - n=2 {1,1}(nem kell tudni általánosan)
Ha $< f \mid g >= \int_{-1}^{1} f[x]\ g[x]\ dx$, akkor OPR elemei az un. Legendre polinomok.

2. Legendre polinom, pl. n=2: 1/ 2 (3x^2-1)

## Határozatlan együtthatók módszere a formulák meghatározozására

Kísérlet a súlyok és az osztópontok meghatározására a pontossági rendre vonatkozó feltételbôl
n=2: osztópontok x[0], x[1]
n=3: osztópontok x[0], x[1], x[2]

```
Clear[W, X, w, x];

W = Table[w[j], {j, 0, 5}];
X = Table[x[j], {j, 0, 5}];

sop[expr_, var_, n_] := Sum[W⟦j⟧ expr /. x → X⟦j⟧, {j, n}]

t1 = Table[sop[x^n, x, 2], {n, 0, 3}]
```

$\{w[0] + w[1],\ w[0]\ x[0] + w[1]\ x[1],\ w[0]\ x[0]^2 + w[1]\ x[1]^2,\ w[0]\ x[0]^3 + w[1]\ x[1]^3\}$

```
t2 = Table[Integrate[x^n, {x, -1, 1}], {n, 0, 3}]
```

$\{2,\ 0,\ \frac{2}{3},\ 0\}$

```
Thread[Equal[t1, t2]]
```

$\{w[0] + w[1] == 2,\ w[0]\ x[0] + w[1]\ x[1] == 0,\ w[0]\ x[0]^2 + w[1]\ x[1]^2 == \frac{2}{3},\ w[0]\ x[0]^3 + w[1]\ x[1]^3 == 0\}$

2n-edfokú nemlineáris egyenletrendszer

```
TableForm[%]
```

$w[0]+w[1]==2$
$w[0]\,x[0]+w[1]\,x[1]==0$
$w[0]\,x[0]^2+w[1]\,x[1]^2==\frac{2}{3}$
$w[0]\,x[0]^3+w[1]\,x[1]^3==0$

```
Solve[Thread[Equal[t1, t2]]]
```

$$\left\{\left\{w[0]\to 1,\ w[1]\to 1,\ x[1]\to -\frac{1}{\sqrt{3}},\ x[0]\to \frac{1}{\sqrt{3}}\right\},\ \left\{w[0]\to 1,\ w[1]\to 1,\ x[1]\to \frac{1}{\sqrt{3}},\ x[0]\to -\frac{1}{\sqrt{3}}\right\}\right\}$$

Levezetett kvadratúraformula:

$Q=f(-\frac{1}{\sqrt{3}})+f(+\frac{1}{\sqrt{3}})$ ([a,b]=[-1,1])

```
t1 = Table[sop[x^n, x, 3], {n, 0, 5}]
```

$$\{w[0]+w[1]+w[2],\ w[0]\,x[0]+w[1]\,x[1]+w[2]\,x[2],\ w[0]\,x[0]^2+w[1]\,x[1]^2+w[2]\,x[2]^2,\ w[0]\,x[0]^3+w[1]\,x[1]^3+w[2]\,x[2]^3,\ w[0]\,x[0]^4+w[1]\,x[1]^4+w[2]\,x[2]^4,\ w[0]\,x[0]^5+w[1]\,x[1]^5+w[2]\,x[2]^5\}$$

```
t2 = Table[Integrate[x^n, {x, -1, 1}], {n, 0, 5}]
```

$$\left\{2,\ 0,\ \frac{2}{3},\ 0,\ \frac{2}{5},\ 0\right\}$$

```
Solve[Thread[Equal[t1, t2]]][[4]]
```

$$\left\{w[0]\to\frac{5}{9},\ w[1]\to\frac{8}{9},\ w[2]\to\frac{5}{9},\ x[2]\to\sqrt{\frac{3}{5}},\ x[1]\to 0,\ x[0]\to-\sqrt{\frac{3}{5}}\right\}$$

Megjegyzés. Az osztópontok permutációtól eltekintve egyértelmû.

Érdekesség n=4. Gröbner bázisok elmélete

```
t1 = Table[sop[x^n, x, 4], {n, 0, 7}];

t2 = Table[Integrate[x^n, {x, -1, 1}], {n, 0, 7}];

AbsoluteTiming[GroebnerBasis[Thread[Plus[t1, -t2]], Join[Take[X, 4], Take[W, 4]]]]
```

$$\{0.138082,\ \{49-216\,w[3]+216\,w[3]^2,\ 49-216\,w[2]+216\,w[2]^2,$$
$$167-216\,w[1]-216\,w[2]+216\,w[1]\,w[2]-216\,w[3]+216\,w[1]\,w[3]+216\,w[2]\,w[3],$$
$$49-216\,w[1]+216\,w[1]^2,\ -2+w[0]+w[1]+w[2]+w[3],\ -51+72\,w[3]+35\,x[3]^2,$$
$$-x[2]+w[2]\,x[2]+w[3]\,x[2]-x[3]+w[2]\,x[3]+w[3]\,x[3],\ -51+72\,w[2]+35\,x[2]^2,$$
$$w[2]\,x[1]-w[3]\,x[1]+w[1]\,x[2]-w[3]\,x[2]+2\,x[3]-w[1]\,x[3]-w[2]\,x[3]-2\,w[3]\,x[3],$$
$$-x[1]+w[1]\,x[1]+w[3]\,x[1]-x[3]+w[1]\,x[3]+w[3]\,x[3],$$
$$123-72\,w[1]-72\,w[2]-72\,w[3]+35\,x[1]\,x[2]+35\,x[1]\,x[3]+35\,x[2]\,x[3],$$
$$-51+72\,w[1]+35\,x[1]^2,\ x[0]+x[1]+x[2]+x[3]\}\}$$

```
Solve[%[[2, 1]] == 0]
```

$\left\{\left\{w[3] \to \frac{1}{36}\left(18 - \sqrt{30}\right)\right\}, \left\{w[3] \to \frac{1}{36}\left(18 + \sqrt{30}\right)\right\}\right\}$

```
N[w[3] /. %]
```

{0.347855, 0.652145}

```
AbsoluteTiming[GroebnerBasis[Thread[Plus[t1, -t2]], Join[Take[W, 4], Take[X, 4]]]]
```

$\{0.038090, \{3 - 30\, x[3]^2 + 35\, x[3]^4, -6\, x[2] + 7\, x[2]^3 - 6\, x[3] + 7\, x[2]^2\, x[3] + 7\, x[2]\, x[3]^2 + 7\, x[3]^3,$
$-6 + 7\, x[1]^2 + 7\, x[1]\, x[2] + 7\, x[2]^2 + 7\, x[1]\, x[3] + 7\, x[2]\, x[3] + 7\, x[3]^2,$
$x[0] + x[1] + x[2] + x[3], -51 + 72\, w[3] + 35\, x[3]^2, -51 + 72\, w[2] + 35\, x[2]^2,$
$-21 + 72\, w[1] - 35\, x[1]\, x[2] - 35\, x[2]^2 - 35\, x[1]\, x[3] - 35\, x[2]\, x[3] - 35\, x[3]^2,$
$-21 + 72\, w[0] + 35\, x[1]\, x[2] + 35\, x[1]\, x[3] + 35\, x[2]\, x[3]\}\}$

```
Solve[%[[2, 1]] == 0]
```

$\left\{\left\{x[3] \to -\sqrt{\frac{1}{35}\left(15 - 2\sqrt{30}\right)}\right\}, \left\{x[3] \to \sqrt{\frac{1}{35}\left(15 - 2\sqrt{30}\right)}\right\},\right.$
$\left.\left\{x[3] \to -\sqrt{\frac{1}{35}\left(15 + 2\sqrt{30}\right)}\right\}, \left\{x[3] \to \sqrt{\frac{1}{35}\left(15 + 2\sqrt{30}\right)}\right\}\right\}$

```
N[x[3] /. %]
```

{-0.339981, 0.339981, -0.861136, 0.861136}

```
t1 = Table[sop[x^n, x, 5], {n, 0, 9}];

t2 = Table[Integrate[x^n, {x, -1, 1}], {n, 0, 9}];
AbsoluteTiming[First[GroebnerBasis[Thread[Plus[t1, -t2]], Join[Take[X, 5], Take[W, 5]]]]]
```

$\{0.961272, -653184 + 5269775\, w[4] - 13005000\, w[4]^2 + 10125000\, w[4]^3\}$

```
Solve[%[[2]] == 0, w[4]]
```

$\left\{\left\{w[4] \to \frac{128}{225}\right\}, \left\{w[4] \to \frac{1}{900}\left(322 - 13\sqrt{70}\right)\right\}, \left\{w[4] \to \frac{1}{900}\left(322 + 13\sqrt{70}\right)\right\}\right\}$

```
N[%]
```

{{w[4.] → 0.568889}, {w[4.] → 0.236927}, {w[4.] → 0.478629}}

```
AbsoluteTiming[First[GroebnerBasis[Thread[Plus[t1, -t2]], Join[Take[W, 5], Take[X, 5]]]]]
```

$\{0.965333, 15\, x[4] - 70\, x[4]^3 + 63\, x[4]^5\}$

```
Solve[%[[2]] == 0, x[4]]
```

$\left\{\{x[4] \to 0\}, \left\{x[4] \to -\frac{1}{3}\sqrt{\frac{1}{7}\left(35 - 2\sqrt{70}\right)}\right\}, \left\{x[4] \to \frac{1}{3}\sqrt{\frac{1}{7}\left(35 - 2\sqrt{70}\right)}\right\},\right.$
$\left.\left\{x[4] \to -\frac{1}{3}\sqrt{\frac{1}{7}\left(35 + 2\sqrt{70}\right)}\right\}, \left\{x[4] \to \frac{1}{3}\sqrt{\frac{1}{7}\left(35 + 2\sqrt{70}\right)}\right\}\right\}$

```
N[%]
```

{{x[4.] → 0.}, {x[4.] → -0.538469}, {x[4.] → 0.538469}, {x[4.] → -0.90618}, {x[4.] → 0.90618}}

- **Kísérlet (határozatlan együtthatók módszere ortogonális polinomok meghatározására): Speciális eset. a=-1,b=1, $\rho$=1, n=3**

```
q[x_] := c0 + c1 x + c2 x^2 + c3 x^3;
```

q ortogonális 1,x,x^2 polinomokra:

$$\left\{\int_{-1}^{1} q[x]\,1\,dx == 0,\ \int_{-1}^{1} q[x]\,x\,dx == 0,\ \int_{-1}^{1} q[x]\,x^2\,dx == 0\right\}$$

$$\left\{2\,c0 + \frac{2\,c2}{3} == 0,\ \frac{2\,c1}{3} + \frac{2\,c3}{5} == 0,\ \frac{2\,c0}{3} + \frac{2\,c2}{5} == 0\right\}$$

```
Solve[%]
```

$$\left\{\left\{c0 \to 0,\ c2 \to 0,\ c1 \to -\frac{3\,c3}{5}\right\}\right\}$$

$$q[x]\ /.\ \left\{\left\{c0 \to 0,\ c2 \to 0,\ c1 \to -\frac{3\,c3}{5}\right\}\right\}\ /.\ \{c3 \to 5/2\}$$

$$\left\{-\frac{3\,x}{2} + \frac{5\,x^3}{2}\right\}$$

Ez a köbös Legendre polinom:

```
LegendreP[3, x]
```

$$\frac{1}{2}\left(-3\,x + 5\,x^3\right)$$

```
? LegendreP
```

LegendreP[$n$, $x$] gives the Legendre polynomial $P_n(x)$.
LegendreP[$n$, $m$, $x$] gives the associated Legendre polynomial $P_n^m(x)$. ≫

```
LegendreP[2, x]
```

$$\frac{1}{2}\left(-1 + 3\,x^2\right)$$

A Legendre polinomok ortogonális rendszert alkotnak
Az ortogonalitás ellenõrzése $< L_1 \mid L_2 > = 0$

```
Integrate[LegendreP[1, x] LegendreP[2, P], {x, -1, 1}]
```

0

```
TableForm[Table[{i, LegendreP[i, x]}, {i, 0, 10}]]
```

| | |
|---|---|
| 0 | $1$ |
| 1 | $x$ |
| 2 | $\frac{1}{2}\left(-1+3x^2\right)$ |
| 3 | $\frac{1}{2}\left(-3x+5x^3\right)$ |
| 4 | $\frac{1}{8}\left(3-30x^2+35x^4\right)$ |
| 5 | $\frac{1}{8}\left(15x-70x^3+63x^5\right)$ |
| 6 | $\frac{1}{16}\left(-5+105x^2-315x^4+231x^6\right)$ |
| 7 | $\frac{1}{16}\left(-35x+315x^3-693x^5+429x^7\right)$ |
| 8 | $\frac{1}{128}\left(35-1260x^2+6930x^4-12012x^6+6435x^8\right)$ |
| 9 | $\frac{1}{128}\left(315x-4620x^3+18018x^5-25740x^7+12155x^9\right)$ |
| 10 | $\frac{1}{256}\left(-63+3465x^2-30030x^4+90090x^6-109395x^8+46189x^{10}\right)$ |

Ortogonalitás

```
TableForm[Table[Integrate[LegendreP[i, x] LegendreP[j, x], {x, -1, 1}], {i, 0, 5}, {j, 0, 5}]]
```

| | | | | | |
|---|---|---|---|---|---|
| 2 | 0 | 0 | 0 | 0 | 0 |
| 0 | $\frac{2}{3}$ | 0 | 0 | 0 | 0 |
| 0 | 0 | $\frac{2}{5}$ | 0 | 0 | 0 |
| 0 | 0 | 0 | $\frac{2}{7}$ | 0 | 0 |
| 0 | 0 | 0 | 0 | $\frac{2}{9}$ | 0 |
| 0 | 0 | 0 | 0 | 0 | $\frac{2}{11}$ |

Mintapontok: a Legendre polinomok gyökei

```
Solve[LegendreP[3, x] == 0]
```

$$\left\{\{x\to 0\},\left\{x\to -\sqrt{\frac{3}{5}}\right\},\left\{x\to \sqrt{\frac{3}{5}}\right\}\right\}$$

```
r3 = Sort[x /. Solve[LegendreP[3, x] == 0], Less]
```

$$\left\{-\sqrt{\frac{3}{5}},\ 0,\ \sqrt{\frac{3}{5}}\right\}$$

A súlyok meghatározása a pontossági rend alapján.

```
plist = {1, x, x^2, x^3, x^4, x^5}; w = {w0, w1, w2};
```

```
Table[
 Sum[w[[i]] (plist[[j]] /. x → r3[[i]]), {i, 3}] == Integrate[plist[[j]], {x, -1, 1}], {j, 3}]
```

$$\left\{w0+w1+w2==2,\ -\sqrt{\frac{3}{5}}\ w0+\sqrt{\frac{3}{5}}\ w2==0,\ \frac{3\ w0}{5}+\frac{3\ w2}{5}==\frac{2}{3}\right\}$$

```
Solve[%]
```

$$\left\{\left\{w0\to\frac{5}{9},\ w2\to\frac{5}{9},\ w1\to\frac{8}{9}\right\}\right\}$$

Itt vannak a súlyok általában $(2/(1-x_i)(P'[x_i])^2)$:

```
Weight[i_, n_, var_] := Module[{xi = Sort[var /. Solve[LegendreP[n, var] == 0], Less][[i + 1]]},
  2/((1 - xi^2) (D[LegendreP[n, var], var] /. var → xi)^2)]
```

```
Simplify[Weight[2, 3, x]]
```

$$\frac{5}{9}$$

```
Simplify[Table[Weight[i, n, x], {n, 5}, {i, 0, n - 1}]]
```

$$\left\{\{2\}, \{1, 1\}, \left\{\frac{5}{9}, \frac{8}{9}, \frac{5}{9}\right\}, \left\{\frac{1}{2} - \frac{\sqrt{\frac{5}{6}}}{6}, \frac{1}{36}\left(18 + \sqrt{30}\right), \frac{1}{36}\left(18 + \sqrt{30}\right), \frac{1}{2} - \frac{\sqrt{\frac{5}{6}}}{6}\right\}, \left\{\frac{1}{900}\left(322 - 13\sqrt{70}\right), \frac{1}{900}\left(322 + 13\sqrt{70}\right), \frac{128}{225}, \frac{1}{900}\left(322 + 13\sqrt{70}\right), \frac{1}{900}\left(322 - 13\sqrt{70}\right)\right\}\right\}$$

- **Kísérlet (osztópontok ismeretében súlyok) Speciális eset. a=-1,b=1, $\rho$=1, n=3**

```
Clear[w]
```

```
W = Table[w[j], {j, 0, 5}];
X = Table[x[j], {j, 0, 5}];
```

```
Map[sop[#, x, 3] &, Table[LegendreP[j, x], {j, 0, 2}]] /.
 {x[0] → -Sqrt[3/5], x[1] → 0, x[2] → Sqrt[3/5]}
```

$$\left\{w[0] + w[1] + w[2], -\sqrt{\frac{3}{5}}\, w[0] + \sqrt{\frac{3}{5}}\, w[2], \frac{2\,w[0]}{5} - \frac{w[1]}{2} + \frac{2\,w[2]}{5}\right\}$$

```
Thread[Equal[%, {2, 0, 0}]]
```

$$\left\{w[0] + w[1] + w[2] == 2, -\sqrt{\frac{3}{5}}\, w[0] + \sqrt{\frac{3}{5}}\, w[2] == 0, \frac{2\,w[0]}{5} - \frac{w[1]}{2} + \frac{2\,w[2]}{5} == 0\right\}$$

```
Solve[%]
```

$$\left\{\left\{w[0] \to \frac{5}{9}, w[2] \to \frac{5}{9}, w[1] \to \frac{8}{9}\right\}\right\}$$

```
Integrate[LegendreP[1, x], {x, -1, 1}]
```

```
Integrate[LegendreP[2, x], {x, -1, 1}]
```

0

```
Manipulate[Plot[{0, LegendreP[n, x]}, {x, -1, 1},
  PlotRange → {-1, 1}, Filling → {1 → {{2}, {Red, Blue}}}], {n, 1, 7, 1}]
```




```
Manipulate[Plot[{0, LegendreP[n, x]}, {x, -1, 1}, PlotRange → {-1, 1},
  Epilog → {Red, PointSize[.02], Point[{x, 0} /. NSolve[LegendreP[n, x] == 0, x]]}], {n, 2, 7, 1}]
```



```
TT = Plot[Table[LegendreP[n, x], {n, 1, 11}], {x, -1, 1}, Epilog → {Red, PointSize[.02], Point[
    Flatten[Table[{x, 0} /. NSolve[LegendreP[n, x] == 0, x], {n, 11}], 1]]}, PlotRange → {-1, 1}]
```




## ■ Feladat

Hasonlítsuk össze a trapézszabályt, az n=2-höz tartozó Gauss kvadratúrát és az integrál pontos értékét, ha , f[x]=x^3+x^2-1

```
f1[x_] := x^3 + x^2 - 1;
```

$$\int_{-1}^{1} f1[x]\, dx$$

$$-\frac{4}{3}$$

```
p = Integrate[f1[x], x]
```

$$-x + \frac{x^3}{3} + \frac{x^4}{4}$$

```
(p /. x → 1) - (p /. x -> -1)
```

$$-\frac{4}{3}$$

```
(2/2) (f1[-1] + f1[1])
```

0

Nem pontos

```
Solve[LegendreP[2, x] == 0]
```

$$\left\{\left\{x \to -\frac{1}{\sqrt{3}}\right\}, \left\{x \to \frac{1}{\sqrt{3}}\right\}\right\}$$

```
1 f1[-1/Sqrt[3]] + 1 f1[1/Sqrt[3]]
```

$$-\frac{4}{3}$$

A Gauss kvadratúra pontos értéket ad, a formula pontossági rendje 3, vagyis köbös polinomokra a hiba 0!

```
f2[x_] := x^4;
```

```
f2[2]
```

16

```
1 f2[-1 / Sqrt[3]] + 1 f2[1 / Sqrt[3]]
```

$\frac{2}{9}$

$$\int_{-1}^{1} \mathrm{f2}[x]\, dx$$

$\frac{2}{5}$

## ■ Gram-Schmidt

```
P[elem_, l_] := Module[{},
  loc = elem - Sum[((Integrate[elem l[[i]], {x, -1, 1}]) / Integrate[l[[i]]^2, {x, -1, 1}]) l[[i]], {i, 1, Length[l]}]; Simplify[(1 / loc /. x → 1) loc]]
```

```
P[x, {1}]
```

x

```
P[x^2, {1, x}]
```

$\frac{1}{2}\left(-1+3x^2\right)$

```
P[x^3, {1, x, 1 / 2 (3 x^2 - 1)}]
```

$\frac{1}{2}x\left(-3+5x^2\right)$

```
LegendreP[3, x]
```

$\frac{1}{2}\left(-3x+5x^3\right)$

```
P[x^4, {1, x, 1 / 2 (3 x^2 - 1), 1/2 (-3 x + 5 x^3)}]
```

$\frac{1}{8}\left(3-30x^2+35x^4\right)$

```
LegendreP[4, x]
```

$\frac{1}{8}\left(3-30x^2+35x^4\right)$

```
L5 = P[x^5, {1, x, 1 / 2 (3 x^2 - 1), 1/2 (-3 x + 5 x^3), 1/8 (3 - 30 x^2 + 35 x^4)}]
```

$$\int_{-1}^{1} \mathrm{L5}\ 1/2\,(3x^2-1)\, dx == 0$$